
1. THE FEMAP API AND OBJECT ORIENTED PROGRAMMING 

1.1 WHAT IS AN API? 

• The FEMAP API is an OLE/COM-based programming interface and is object oriented programming. If you have 
never programmed in an object oriented code, it can seem quite different and foreign.  

• API means “Application Programming Interface”. It is important to understand that the API script you write is not 
part of FEMAP, but is a stand-alone program that is interacting with FEMAP. 

• There are a number of codes that can call FEMAP through the API: Visual Basic, VBA (Excel, Word, Access, etc.), C, 
or C++.  

• The most commonly used codes used are Visual Basic, VBA, and WinWrap. 

• WinWrap is a flavor of Visual Basic that is included with FEMAP. In the FEMAP interface, WinWrap is 
noncompilable, for this reason many choose not to use it, but it is a very convenient way to program if your 
specific application does not need to be compiled.  

• This seminar will focus on using WinWrap via the FEMAP API window. 

This is the optional FEMAP API editing window. Although the window appears to be part of your FEMAP session, it is not. 
It is merely a code editing tool. 

1.2 WHAT IS OBJECT ORIENTED PROGRAMMING? 

Traditional programming is usually seen as being a set of functions, or simply as a list of instructions. 

• Object Oriented Programming (or OOP) can be seen as a group of objects that cooperate with each other. Each of 
the objects has their own distinct set of capabilities. 

 

It is helpful to think of each of the entities as being separate. 

• Your Visual Basic code acts like a traditional code, i.e. as a set of instructions. 

• The VB code makes requests of the API, which then acts upon those requests either by retrieving from and putting 
things into the FEMAP database. 

• Remember, FEMAP is a database, which only holds and displays data. 

1.3 FEMAP API OBJECTS 

The objects found in the FEMAP API fall into three categories: 

• FEMAP Application Object. This is the “king of the objects” within the FEMAP API. It is needed to create other 
objects. It is also the object that will be used to perform operations, measure, calculate, display messages, etc. 

o For example, create the FEMAP Application Object: 
Dim App As femap.model 

Set App = feFemap() 

o Then using the feNode Method, the FEMAP Application Object can create a Node Object: 
Dim nd As femap.Node 

Set nd = App.feNode 

 

• FEMAP Tool Objects. These objects allow access to the various windows available in the user interface, others 
provide entity selection, and general data and file functionality. The Set Object is one of the most common. 

o Use the feSet Method of the FEMAP Application Object to create Set objects: 
Dim ndSet As femap.Set 

Set ndSet = App.feSet 

o Then use the Select Method of the Set Object to prompt the user to pick nodes: 
ndSet.Select(FT_NODE, True, "Pick Nodes") 

 

• FEMAP Entity Objects. Objects. For every entity within FEMAP (nodes, elements materials, loads, etc.) there is a 
corresponding object within the API. These objects are used to manipulate entities. 

o The Node Object can now manipulate the coordinates of nodes: 
nd.x = 100 

nd.y = 26.1 

nd.z = 5.23 

2. INTRODUCTION TO FEMAP’S API 

At the end of this seminar, we will walk through an existing 
program, line-by-line, and identify the building blocks of an 
API. 

That can be a bit boring, so before we get to that, let’s briefly 
talk about some of the key concepts for working with the 
FEMAP API: 

 

• FEMAP Application Object 

• Set Object 

• Object Methods 

• Object Properties 

• Variables 

• Data Types 

• Return Codes 

2.1  “HELLO WORLD” – YOUR FIRST FEMAP API 

Programming can be very intimidating. For the uninitiated, the code sitting within a FEMAP API file is confusing and 
overwhelming. In this example, we will take the first step in programming and simply prove to ourselves that connection to 
the FEMAP interface through the API can be accomplished by “non-programmers”.  

Topics Covered: 

• API Files vs. Program Files 

• Connecting to FEMAP 

• Auto Complete within the API Programming Window 

• Data Types 

• Parameter Info 

• Hunting the FEMAP help Files  

 

Dimensioning the FEMAP Application Object, see 
FEMAP API Help Section  Section 3.1.2: 

Method 1** 

Dim App As femap.model 

Set App = feFemap() 

Method 2* 

Dim App As femap.model 

Set App = GetObject(,"femap.model") 

Method 3 

Dim App As Object 

Set App = GetObject(, "femap.model") 

**(Does not work from VB in Excel) 
*(Does not automatically work from VB in Excel) 

3. COLLECTING INFORMATION FROM THE USER 

If you’re just interested in automating a repetitive task without a lot of user interaction, you might be better of using a 
Program File (recorded macro). However, if you need some user interaction, you’ll need a way to collect information 
from them. 

3.1 STANDARD DIALOG BOXES 

Summoning the “Standard Dialog Boxes” involves a variety of FEMAP Application Object Methods to define coordinates, 
vectors, planes or other types of information. These dialog boxes are very useful, nonspecialized methods to prompt the 
user for information. 

 

App.feCoordPick("Select the Location for the Circle Center", PointXYZ) 

 
App.feVectorPick("Select Vector to Move Along", False, vecLength, vecBase, vecDir) 

 
App.fePlanePick("Define the Cutting Plane", plBase, plNormal, plAxis) 

3.2 SET OBJECTS 

Set objects are the most common of the tool objects and any program that requires the user to select entities is going to 
utilize Set objects. 

  

The Select Method brings up the standard entity selection 
dialog box with all the bells and whistles to facilitate picking 
what you want. 

entSET.Select(FT_ELEM, True, "Select Element(s)") 

 

The SelectMultiID Method brings up dialog box allowing the 
user to select entities from a list. In most cases you should 
only use it for entities with titles. 

entSET.SelectMultiID(FT_PROP, 1, "Select Properties") 

 

 

 

 

The SelectID Method brings up an entity selection dialog box 
that only allows one entity to be picked. 

entSET.SelectID(FT_NODE, "Select One Node", ndID) 

4. WORKING WITH FEMAP ENTITIES 

Incorporating materials and properties into our models is really where we start to differentiate FEA from CAD. It might 
seem like there isn’t a lot of time to be saved by automating the creation or modification of materials and properties, but 
the subsequent programs will show that that these skills are important to have in your programming toolbox. 

4.1 ACCESSING MATERIAL PROPERTIES 

Here’s a guide to accessing the various properties of an isotropic material card. 

 

 

Sub Main 

 Dim App As femap.model 

 Set App = feFemap() 

 

 Dim Mat As femap.Matl 

 Set Mat = App.feMatl 

 

 Dim MatSet As femap.Set 

 Set MatSet = App.feSet 

 Dim MatID As Long 

 

 MatSet.Select(FT_MATL, True, "Select Materials") 

 MatID = MatSet.First 

 

 While MatID > 0 

  Mat.Get(MatID) 

  App.feAppMessage(4, "Mat ID = " + CStr(MatID)) 

  App.feAppMessage(2, "E = " + CStr(Mat.mval(0))) 

  App.feAppMessage(2, "G = " + CStr(Mat.mval(3))) 

  App.feAppMessage(2, "v = " + CStr(Mat.mval(6))) 

  App.feAppMessage(2, "CTE = " + CStr(Mat.mval(36))) 

  App.feAppMessage(2, "k = " + CStr(Mat.mval(42))) 

  App.feAppMessage(2, "Cp = " + CStr(Mat.mval(48))) 

  App.feAppMessage(2, "Density = " + CStr(Mat.mval(49))) 

  App.feAppMessage(2, "Damping = " + CStr(Mat.mval(50))) 

  App.feAppMessage(2, "Temp = " + CStr(Mat.mval(51))) 

  App.feAppMessage(2, "Tens = " + CStr(Mat.mval(52))) 

  App.feAppMessage(2, "Comp = " + CStr(Mat.mval(54))) 

  App.feAppMessage(2, "Shear = " + CStr(Mat.mval(56))) 

  App.feAppMessage(2, "Heat = " + CStr(Mat.mval(100))) 

 

  MatID = MatSet.Next 

 Wend 

End Sub 

4.2 ACCESSING ELEMENT PROPERTIES 

Fortunately, element Properties aren’t as complicated and numerous as property or material Properties. That is, the 
majority of the data is held in the FEMAP materials and properties. The most complicated part of defining an element is 
managing the nodes. 

Let’s take a look at some of the more straightforward properties like ID, color, layer, etc.: 

 

 

    elemOBJ.ID = 19 

    elemOBJ.type = FET_L_BEAM 

    elemOBJ.topology = FTO_LINE2 

    elemOBJ.color = 31508 

    elemOBJ.layer = 1 

    elemOBJ.propID = 6 

    elemOBJ.Node(0) = 46659 

    elemOBJ.Node(1) = 128139 

    elemOBJ.release(0, 0) = 1 

    elemOBJ.release(0, 1) = 1 

    elemOBJ.release(0, 2) = 1 

    elemOBJ.orient(0) = 0 

    elemOBJ.orient(1) = 0 

    elemOBJ.orient(2) = 1 

4.3 ACCESSING VIEW PROPERTIES 

This can be a bit confusing because the syntax of View Object Properties differs slightly from other object Properties. 
Let’s look at a few examples 

 

 
 

Sub Main 

    Dim App As femap.model 

    Set App = feFemap() 

 

    Dim viewobj As femap.View 

    Set viewobj = App.feView 

 

    Dim viewID As Long 

    App.feAppGetActiveView(viewID) 

 

    viewobj.Get(viewID) 

    viewobj.Draw(0) = False 

    viewobj.Label(0) = 73 

    viewobj.ColorMode(0) = 1 

    viewobj.color(0) = 1 

    viewobj.RenderPushLabel = (1) 

    viewobj.Put(viewID) 

End Sub 

5. CONTROLLING PROGRAM FLOW 

5.1 LOOPS AND IF-THEN STATEMENTS 

Loops are commonly used to work through entities selected by a set object. “For” loops continue for a specified number 
of iterations. “While” loops check a condition at the beginning of the loop and continue until the condition is false. 

The standard If-Then statement is a great way to filter entities (If the element is a beam, then list ID number) or control 
the flow of the program (If rc is equal to 0, Then exit the program). Incorporating Else and ElseIf statements simply opens 
up more flow options, reducing the need for multiple nested If-Then statements. Let take a look at some examples 

Check to see if “nodeID” is greater than zero. If so, 
perform the specified actions. Repeat until nodeID ≯ 0. 

If the background is black (0) then switch it to white (124), 
otherwise switch it to black (0). 

 

Sub Main 

 Dim App As femap.model 

 Set App = feFemap() 

 

 Dim nodeOBJ As femap.Node 

 Set nodeOBJ = App.feNode 

 

 Dim nodeSET As femap.Set 

 Set nodeSET = App.feSet 

 Dim nodeID As Long 

 

 nodeSET.Select(FT_NODE, True, "Select Nodes") 

 

 nodeID = nodeSET.First 

 While nodeID > 0 

  nodeOBJ.Get(nodeID) 

  App.feAppMessage(2,CStr(nodeOBJ.x)) 

  App.feAppMessage(2,CStr(nodeOBJ.y)) 

  App.feAppMessage(2,CStr(nodeOBJ.z)) 

  nodeID = nodeSET.Next 

 Wend 

 

End Sub 

 

Sub Main 

 Dim App As femap.model 

 Set App=feFemap() 

 

 Dim vobj As femap.View 

 Set vobj=App.feView 

 

 Dim viewID As Long 

 

 App.feAppGetActiveView(viewID) 

 vobj.Get(viewID) 

 

 If vobj.WindowBackColor=0 Then 

  vobj.WindowBackColor=124 

 Else 

  vobj.WindowBackColor=0 

 End If 

 

 vobj.Put(viewID) 

 App.feViewRegenerate(0) 

 

End Sub 

6. CONNECTING TO FEMAP AND EXCEL 

One of the abilities that makes the FEMAP API so powerful is the connection to MS Excel or other external programs. 
Whether you’re dumping results to a spreadsheet or pushing a massive amount of load data into your model, connecting to 
external databases is a game changer.  

Topics Covered: 

• Connecting to FEMAP from Excel 

• Printing information to an Excel Workbook 

• Pulling information from an Excel Workbook 
 

Workflow: 

• Open a new Excel worksheet 

• Activate the Developer Tab 

• Open the Visual Basic Editor 

• Import the program from Part 1 

• Update the program to get it to cooperate 
with Excel 

Remember this? Why is it important? 

Method 1** 

Dim App As femap.model 

Set App = feFemap() 

• Does not work from VB in Excel. 

• Provides autocomplete to assist with programming 

• Ensures connection to current FEMAP session  

Method 2* 

Dim App As femap.model 

Set App = GetObject(,"femap.model") 

• Requires the user to add the Femap Type Library to 
the Excel Workbook References.  

• Provides autocomplete to assist with programming 

Method 3 

Dim App As Object 

Set App = GetObject(, "femap.model") 

• Does not require special setup in Excel (References)  

• There’s no autocomplete  

 

Important information from the FEMAP help file:   

“When you are connecting to an existing session there are some special considerations. If you are programming in an 
environment outside of FEMAP (anything but the FEMAP API Programming Window), then you must use “GetObject” to 
connect. This uses standard OLE/COM mechanisms to find the FEMAP object, but imposes a limitation. If you have multiple 
copies of FEMAP running (multiple processes, not multiple models open in the same FEMAP), GetObject will always connect 
you to the process that was started first. This is simply a limitation of the OLE/COM interface and can not be avoided. 

If you are using the integrated WinWrap environment (API Programming) however, you can overcome this limitation by 
never using “GetObject”. Instead, you can connect to the current FEMAP session, no matter how many FEMAP processes 
are running by using” 

Excel > Developer > Visual Basic > Insert > Module 

From the Developer tab of Excel, click on the Visual Basic icon. Once 

the Visual Basic window opens, click Insert > Module. Paste the 

program into the Module. You can now close Visual Basic. 

Excel > Developer > Insert > Form Controls 

Insert a button into the spreadsheet. Click and drag to set the button 
size. When you let go of the mouse button, an “Assign Macro” dialog 
box will appear. Select the LoadNodalData program that you just 
created in Visual Basic. The button is now linked to the program and 
is fully functional. Rename the button by right clicking on the button 
and modifying the text. 

6.1 SHOW AND TELL: PRESSURE VESSEL PIPING LOAD APPLICATOR 

Years ago, I had a pressure vessel project…THE pressure vessel project. From the geometry, to the loads, to the 
documentation requirements, the complexity was endless. One example of a mind-numbing task was applying the 
internal piping loads. There were eight internal vessels (pulse jet mixers, PJMs) within the parent vessel and it was 
possible for each one to put load on the internal piping of the vessel. For each pipe, the load acted on a vector from the 
centerline of the PJM to the centerline of the pipe. The magnitude of the load was a function of height and distance from 
the PJM. It was brutal and the FEMAP API was my saving grace. 

  

The first step was extracting nodes from the FEMAP model. We needed node ID number and coordinates. This 
information was used in combination to calculate the direction and magnitude of the load. The next step was to push the 
loads calculated in the Excel workbook, back into the FEMAP database. The process takes about 30 seconds with the API. 
Applying the loads manually takes about 8 hours.  

Here’s the workflow: 

CLICK THE “GATHER NODAL DATA” BUTTON, SWITCH TO THE FEMAP WINDOW 

When you switch to the FEMAP window, you will see a nodal entity selection dialog box waiting for you. Select some of the pink nodes 
on the beam elements. The color of the nodes centered between supports has been modified to make them easier to find. To select all 
of the pink nodes, use the “Color” selection method. 

 

SWITCH TO THE EXCEL SPREADSHEET 

Data has been written into the spread sheet for the nodes you selected in FEMAP. 

ENTER ADDITIONAL DATA INTO EXCEL 

The spread sheet requires a pipe length and pipe diameter to calculate loads. For this particular model, the pipe is 170” by 2.2”. Now 
that the spreadsheet has created loads we will use another program to apply them to the FEA model. 

 

CLICK THE BUTTON, SWITCH TO THE FEMAP WINDOW 

The API has updated the FEMAP database with the loads from Excel but they might not show up in the Model Info Tree. Regenerating 
the image (Window > Regenerate, Ctrl G) will update the Model Info Tree with the new loads.  

7. POST PROCESSING 

7.1 FREE BODY DIAGRAM GENERATOR 

Freebody Diagrams (or FBDs) are an advanced post technique, yet they are beautifully simple at the same time. It’s all about 
summing the forces and moments for a selection of elements and nodes. The trick is carefully selecting the entities and 
making sure that the “Freebody Contributions” are logical. “Section Cut” display mode operates in the same manner as the 
Interface Load, but rather than selecting elements and nodes, the user simply selects a cutting plane. FEMAP will 
automatically select nodes along the cutting plane and elements on one side of the plane. 

Topics Covered: 

• FBD Object Properties and Methods 
 

Workflow: 

• Prompt the user to select a vector (this will 
be the normal vector of the section cut 
plane) 

• Print the vector information to Excel 

• Dimension a FBD Object 

• Hardcode some of the FBD Object Properties 

• Define other FBD Object Properties using 
information from Excel 

• Put the FBDs into FEMAP database 

• Extract FBD force data from the FEMAP data 
and print to Excel 

7.2 FREEBODY PROPERTIES 

The most common problems with freebody diagrams are selecting entities and choosing the correct contributions. Using 
the Section Cut display mode eliminates the former while hard coding FBD options into an API eliminates the later.  

8. ANATOMY OF A SIMPLE API 

Sub Main 

 

Dim App As Object 

Set App = GetObject(,"femap.model") 

 

Dim entitySet As Object 

Set entitySet = App.feSet 

 

Dim entityType As Long 

entityType = 7 

 

Dim clearSet As Boolean 

clearSet = True 

 

Dim messageString As String 

messageString = "Please Select the Nodes You Would Like To Move” 

 

rc = entitySet.Select(entityType, clearSet, messageString) 

 

Dim setID As Long 

setID = entitySet.ID 

 

Dim vecMove(3) As Double 

vecMove(0) = 10 

vecMove(1) = 0 

vecMove(2) = 0 

 

Dim vecLength As Double 

rc = App.feVectorLength(vecMove, vecLength) 

 

rc = App.feMoveBy(entityType, setID, False, vecLength, vecMove) 

 

End Sub 

Now we’ll walk through a simple API. All this API 
does is move a set of selected nodes 10 units in the 
x direction.  

 

Yes, there is a function that will do this directly 
without an API, but we are starting simple. The 
entire script is shown on the left. We will walk 
through each step in this API. 

8.1 DEFINING AN OBJECT 
Sub Main 

 

Dim App As Object 

Set App = GetObject(,"femap.model") 

 

Dim entitySet As Object 

Set entitySet = App.feSet 

 

Dim entityType As Long 

entityType = 7 

 

Dim clearSet As Boolean 

clearSet = True 

 

Dim messageString As String 

messageString = "Please Select the Nodes You Would Like To Move” 

 

rc = entitySet.Select(entityType, clearSet, messageString) 

 

Dim setID As Long 

setID = entitySet.ID 

 

Dim vecMove(3) As Double 

vecMove(0) = 10 

vecMove(1) = 0 

vecMove(2) = 0 

 

Dim vecLength As Double 

rc = App.feVectorLength(vecMove, vecLength) 

 

rc = App.feMoveBy(entityType, setID, False, vecLength, vecMove) 

 

End Sub 

Sub Main at the top of the script signifies that what 
follows is the main program. 

 

Next we create an object called “App”. We then set 
this object equal to the current FEMAP session. 
Essentially, this creates the FEMAP Application 
Object and calls it App. So the object App now has 
all the Properties of the FEMAP Application Object. 

 

What we want to do will also require the help of 
another object, the Set Object. In this program, we 
are calling it “entitySet” and creating it using the 
feSet Method. entitySet now has all the Properties 
and Methods inherent in the Set Object. 

8.2 DATA TYPES 

Visual Basic requires the programmer to declare all variables before they are used as well as what type of data they will be. 
The six data types are shown below. WinWrap corresponds to the Visual Basic 6 data types. This table can be found in 
FEMAP API Help Section 1.2 “Data Types”. 

8.3 DIMENSIONING VARIABLES 
Sub Main 

 

Dim App As Object 

Set App = GetObject(,"femap.model") 

 

Dim entitySet As Object 

Set entitySet = App.feSet 

 

Dim entityType As Long 

entityType = 7 

 

Dim clearSet As Boolean 

clearSet = True 

 

Dim messageString As String 

messageString = "Please Select the Nodes You Would Like To Move” 

 

rc = entitySet.Select(entityType, clearSet, messageString) 

 

Dim setID As Long 

setID = entitySet.ID 

 

Dim vecMove(3) As Double 

vecMove(0) = 10 

vecMove(1) = 0 

vecMove(2) = 0 

 

Dim vecLength As Double 

rc = App.feVectorLength(vecMove, vecLength) 

 

rc = App.feMoveBy(entityType, setID, False, vecLength, vecMove) 

 

End Sub 

Next we declare all the variables needed Select 
Method of the Set Object (this will be described in 
more detail on subsequent pages). 

 

First, we declare a variable called “entityType” as a 
4-byte integer data type (Long) and assign it a value. 
This will be used to specify what type of entity we 
will select with the Select Method. 

Then, we declare a variable called “clearSet” as a 
true/false value data type (Boolean) and assign it a 
value. This will be used to indicate that we want the 
Set Object cleared of any previous entities. 

Finally, we declare a variable called “messageString” 
as a character string data type (String) and assign it 
a value. This will be the message the user sees when 
the Set Object uses the Select Method. 

8.4 ENTITY TYPES 

Each entity in the FEMAP API is identified by a name 
and a number. The entity can be referred to by 
either. In the preceding piece of code where I refer 
to the node entity as the number 7, I could also 
have referred to it as FT_NODE. Either way the API 
will know to which entity type you are referring. 

8.5 USING THE CAPABILITIES OF AN OBJECT 
Sub Main 

 

Dim App As Object 

Set App = GetObject(,"femap.model") 

 

Dim entitySet As Object 

Set entitySet = App.feSet 

 

Dim entityType As Long 

entityType = 7 

 

Dim clearSet As Boolean 

clearSet = True 

 

Dim messageString As String 

messageString = "Please Select the Nodes You Would Like To Move” 

 

rc = entitySet.Select(entityType, clearSet, messageString) 

 

Dim setID As Long 

setID = entitySet.ID 

 

Dim vecMove(3) As Double 

vecMove(0) = 10 

vecMove(1) = 0 

vecMove(2) = 0 

 

Dim vecLength As Double 

rc = App.feVectorLength(vecMove, vecLength) 

 

rc = App.feMoveBy(entityType, setID, False, vecLength, vecMove) 

 

End Sub 

        

 

Now that we have defined all of the variables that 
will be used as inputs, we want to do is collect from 
the user – what nodes they would like moved. 

 

The Set Object has a handy Method that allows us 
to do this called Select. 

  

Objects have two types of capabilities: 

• Methods (Functionality) 

• Properties (Data) 



 

 

 

Automation and Optimization with the FEMAP API 
 

AppliedCAx / Predictive Engineering - Please share with your friends and visit us online at www.AppliedCAx.com Page 34 of 42 

8.6 OBJECT METHOD SYNTAX 

 

The syntax in the above statement is standard. The object is what is being “asked” to act. The Method is what the object 
is being asked to do. The inputs are what the object needs in order to execute the Method. The output is what the object 
will “produce”, although often times, Methods will have no output. 

The term rc is the return code and will generate a specific value depending on a number of object success states.  

For more information, see FEMAP API Help Section 5.1.2, “Common Entity Methods”. 

 

8.7 RETURN CODES 

Often statements like the following are found in 
API’s:  

 

rc = object.Method(inputs,output) 

  

The rc stands for return code. After the object 
executes its Method, it returns a code that 
corresponds to its success in executing the Method. 
If the object is successful, a -1 is returned. If it is not 
successful, something else will be returned 
depending upon what went wrong. All the return 
codes are found in the table on the right. 

Methods that Produce Output 

How this all works is best explained by a more concrete example. Think of an object as a person, a person who will do 
things that you ask. We will call this person “Mike”. Say we want “Mike” to go to the store and buy an apple. In order for 
“Mike” to do this, we need to provide him with a car and money. For this Method, “Mike” will produce an output: an 
apple. The statement would look like this: 

 

 

Methods that Produce No Output 

Now suppose we want “Mike” to wash the dishes in the kitchen. We need to provide him with the dishes, soap, a 
sponge, and a sink. After he is done he will produce NO output for us because we haven’t asked him to bring us anything. 
All we have done is ask him to go off and do something. The statement looks much like the previous one. 

Sometimes we ask objects to organize things. Sometimes we will ask them to create or move things. The only time 
objects will have output is if we ask them to bring us something specific. This most likely seems fairly abstract, but once 
you see how it actually works you will see that it is very intuitive.  

 

 

 

Dim entityType As Long 

entityType = 7 

 

Dim clearSet As Boolean 

clearSet = True 

 

Dim messageString As String 

messageString = "Please Select the Nodes You Would Like To Move” 

 

rc = entitySet.Select(entityType, clearSet, messageString) 

 

The syntax of the entitySet. Select object Method 
follows the standard syntax. For this object there is 
no output, only inputs. This is because we are not 
“asking” the object for anything concrete (like a 
value); we are asking the object to place certain 

entities into a set. The effect is having our desired 
entities added to the entitySet. (Later, we will use an 
object that will produce an actual output, a required 

distance.) 

A Set Object is used to store a set of entities, i.e. a list of nodes. The Select Method displays the above shown dialog box 
so the user can select which nodes they are interested in. After the user selects these nodes, they are added to the set 
called entitySet.  

In order to do this, the Set Object needs: 

• to know what type of entity to ask for: entityType, which has already been set to 7 (corresponding to nodes), 

• to know if the object should be cleared of any previous entities: clearSet, has already been set to True, 

• to know what message to display at the top of the entity selection dialog: messageString has already been set to 
“Please Select the Nodes You Would Like To Move”  

8.8 OBJECT PROPERTY SYNTAX 

Sub Main 

 

Dim App As Object 

Set App = GetObject(,"femap.model") 

 

Dim entitySet As Object 

Set entitySet = App.feSet 

 

Dim entityType As Long 

entityType = 7 

 

Dim clearSet As Boolean 

clearSet = True 

 

Dim messageString As String 

messageString = "Please Select the Nodes You Would Like To Move” 

 

rc = entitySet.Select(entityType, clearSet, messageString) 

 

Dim setID As Long 

setID = entitySet.ID 

 

Dim vecMove(3) As Double 

vecMove(0) = 10 

vecMove(1) = 0 

vecMove(2) = 0 

 

Dim vecLength As Double 

rc = App.feVectorLength(vecMove, vecLength) 

 

rc = App.feMoveBy(entityType, setID, False, vecLength, vecMove) 

 

End Sub 

Certain object capabilities require no input and do 
not provide output in the convectional way. These 
are called Properties. 

Such is the case with the entitySet.ID statement. 

 

Instead this syntax returns the desired value to the 
variable on the left hand side of the equal sign. In 
this case setID will take on the ID number of the 
object entitySet. A single program can have multiple 
set objects defined, each containing their own data. 
Each of these sets would have a specific ID to 
differentiate them. 

Note: It's recommended that you never hard code 
set IDs in your program as it may have unintended 
consequences. When a Set Object is created, 
FEMAP will auto-assign the next available ID 
automatically. To reference that set later, simply 
use the ID Property rather than a hard-coded ID. 

 

For more information, see FEMAP API Help Section 
5.1.1, Common Entity Properties. 

Sub Main 

 

Dim App As Object 

Set App = GetObject(,"femap.model") 

 

Dim entitySet As Object 

Set entitySet = App.feSet 

 

Dim entityType As Long 

entityType = 7 

 

Dim clearSet As Boolean 

clearSet = True 

 

Dim messageString As String 

messageString = "Please Select the Nodes You Would Like To Move” 

 

rc = entitySet.Select(entityType, clearSet, messageString) 

 

Dim setID As Long 

setID = entitySet.ID 

 

Dim vecMove(3) As Double 

vecMove(0) = 10 

vecMove(1) = 0 

vecMove(2) = 0 

 

Dim vecLength As Double 

rc = App.feVectorLength(vecMove, vecLength) 

 

rc = App.feMoveBy(entityType, setID, False, vecLength, vecMove) 

 

End Sub 

We will now use the feVectorLength Method of the 
FEMAP Application Object to find the magnitude of 
the nodal move we will be requesting. 

First we declare a 3 dimensional array, composed of 
8-byte real numbers called “vecMove”. This array 
will represent the vector along which the translation 
will take place. We then specify each value in the 
array. This will be the feVectorLength Method 
input. 

Then we declare a variable called “vecLength” as an 
8-byte real number (Double). No value is assigned 
because it will be used for the feVectorLength 
Method output. 

 

 

 

What we are asking of the FEMAP Application 
Object is for it to take our vector, called vecMove, 
and tell us how long it is. What the object gives us is 
a new value for vecLength. If the operation is 
successful, rc will be given the value of –1. 

Sub Main 

 

Dim App As Object 

Set App = GetObject(,"femap.model") 

 

Dim entitySet As Object 

Set entitySet = App.feSet 

 

Dim entityType As Long 

entityType = 7 

 

Dim clearSet As Boolean 

clearSet = True 

 

Dim messageString As String 

messageString = "Please Select the Nodes You Would Like To Move” 

 

rc = entitySet.Select(entityType, clearSet, messageString) 

 

Dim setID As Long 

setID = entitySet.ID 

 

Dim vecMove(3) As Double 

vecMove(0) = 10 

vecMove(1) = 0 

vecMove(2) = 0 

 

Dim vecLength As Double 

rc = App.feVectorLength(vecMove, vecLength) 

 

rc = App.feMoveBy(entityType, setID, False, vecLength, vecMove) 

 

End Sub 

And last, but certainly not least, we will request that 
the FEMAP Application Object moves our nodes. 

This Method, called feMoveBy, has the following 
inputs: 

• what type of entity it is moving, 

• what set contains the ID’s of the entities to 
move, 

• whether or not this is a radial translation, 

• the length of the translation, 

• a vector specifying the direction of the 
translation. 

8.9 CONCLUSION 

What is most interesting about the script we just explored is that it only does one thing: it moves the nodes. Everything 

else found in script exists only to provide the last command with the information it needs to make the move. This is fairly 

common. Often much of the API script is devoted to retrieving things from the database, interpreting them, changing 

them, and then finally inserting them back in. 

In our case, we retrieved the node numbers of that were to be moved, organized them into a set, and then requested 

that the FEMAP Application Object move them. 

The previous example is a simple one that uses very little logic. There are no for or while loops and no if then statements, 

but all of the standard logic statements are available and are used frequently. Anyone with basic programming skills 

should be able to utilize them as they would in any other language. 

You should now understand the basics needed to read and understand basic API’s. The only way to become a pro at 

writing them is to sit down and do it. In no time you will find that the structure and capabilities are extremely powerful. 

You will also find that you will never again need to scratch your head and say, “I wish the FEMAP programmers would 

have included this feature.” 

9. REMEMBER, YOU’RE NOT ALONE! 

Reach out to us at www.AppliedCAx.com for APIs, Technical Seminars and more information on FEMAP and NX Nastran. 
For more information on API programming with FEMAP, check out our online course. 

 

 

 

 

 

 

 

And thanks https://xkcd.com/ for all the great comics! 

